

---

<b>文件字体规则</b> .....	1
<b>系统设定指令</b> .....	2
1. SIZE.....	2
2. CASHDRAWER.....	2
3. GAP.....	3
4. OFFSET.....	4
5. SPEED.....	4
6. DENSITY.....	5
7. DIRECTION.....	5
8. REFERENCE.....	5
9. SHIFT.....	6
10. CODEPAGE.....	6
11. CLS.....	7
12. FEED.....	7
13. BACKFEED&BACKUP.....	8
14. FORMFEED.....	8
15. HOME.....	9
16. PRINT.....	9
17. SOUND.....	10
18. LIMITFEED.....	10
19. SELFTTEST.....	11
<b>卷标内容设计指令</b> .....	11
20. BAR.....	11
21. BARCODE.....	12
22. BOX.....	15
23. BITMAP.....	16
24. PUTBMP.....	18
25. PUTPCX.....	19
26. ERASE.....	20
27. REVERSE.....	20
28. TEXT.....	21
<b>询问打印机状态指令</b> .....	22
29. <ESC>!?.....	22
30. <ESC>!R.....	22
31. ~!@.....	23
32. ~!A.....	23
33. ~!D.....	23
34. ~!F.....	24
35. ~!I.....	24
36. ~!T.....	24
<b>档案管理指令</b> .....	25

37.	DOWNLOAD.....	25
38.	EOP.....	26
39.	FILES.....	27
40.	KILL.....	27
41.	MOVE.....	28
42.	RUN.....	28
<b>BASIC</b>	<b>指令及函式.....</b>	<b>29</b>
43.	ABS().....	29
44.	ASC().....	29
45.	CHR\$().....	30
46.	END.....	31
47.	EOF().....	31
48.	OPEN.....	32
49.	READ.....	34
50.	SEEK.....	35
51.	LOF().....	36
52.	FREAD\$().....	37
53.	FOR...NEXT LOOP.....	38
54.	IF...THEN...ELSE.....	39
55.	GOSUB...RETURN.....	40
56.	GOTO.....	41
57.	REM.....	42
58.	INT().....	43
59.	LEFT\$().....	43
60.	LEN().....	44
61.	MID\$().....	45
62.	RIGHT\$().....	45
63.	STR\$().....	46
64.	VAL().....	47
65.	BEEP.....	48
	<b>打印机外围功能设定指令.....</b>	<b>49</b>
66.	SET COUNTER.....	49
67.	SET KEY1, SET KEY2.....	50
68.	SET PEEL.....	51
69.	SET TEAR & SET STRIPPER.....	51
70.	SET HEAD.....	52
71.	SET COM1.....	53
72.	SET PRINTKEY.....	54
73.	SET REPRINT.....	55
74.	PEEL.....	56
75.	KEY1,KEY2.....	57
	<b>打印机全域变数.....</b>	<b>58</b>
76.	@LABEL.....	58

# 文件字体规则

本文件使用以下字体规则

文件规则	描述
[表示内容]	在中括号的内容表示该参数为选项
<ESC>	<ESC>代表ASCII 27字符，当打印机收到以该控制字符为起始的指令将立即响应（即使打印机在错误状态时也将实时回应）
~	（ASCII 126）,该字符起始的指令用于询问打印机的状态
注： <i>200DPI: 1 mm=8dots</i>	粗斜体Times New Roman字型,用于表示批注
DOWNLOAD "TEST.BAS" SET COUNTER @1 1 @1="0001" TEXT 10,10,"3",0,1,1,@1 PRINT 3,2 EOP	当所列出的内容为程序时以宋体10号字型表示

# 系统设定指令

## 1. SIZE

该指令用于设定卷标纸的宽度及长度


指令语法

(1) 英制系统(inch)

SIZE m, n

(2) 公制系统(mm)

SIZE m mm, n mm



参 数

说 明

m

标签纸的宽度（不含背纸）

n

标签纸的长度（不含背纸）

**Note:** 200 DPI: 1 mm = 8 dots

范例

(1) 英制系统 (inch)

SIZE 3.5, 3.00

(2) 公制系统 (mm)

SIZE 100 mm, 100 mm

## 2. CASHDRAWER

产生钱箱控制脉冲

指令语法

CASHDRAWER m, t1, t2

ESC p m, t1, t2

参 数

说 明

m

0, 48 钱箱插座的引脚2

1, 49 钱箱插座的引脚5

t1 , t2

$0 \leq t1 \leq 255, 0 \leq t2 \leq 255$

输出由t1和t2设定的钱箱开启脉冲到由m指定的引脚

**注释:**

- 1) 钱箱开启脉冲高电平时间为 [t1 x 2ms],低电平时间为 [t2 x 2ms].
- 2) 如果 t2 < t1,低电平时间为 [ t1 x 2ms].

### 3. GAP

该指令定义两张卷标纸间的垂直间距距离

指令语法

(1)英制系统 (inch)

GAP m, n

(2) 公制系统 (mm)

GAP m mm, n mm

参 数

说 明

m


两标签纸中间的垂直距离

$0 \leq m \leq 1$  (inch) ,  $0 \leq m \leq 25.4$  (mm)

n

垂直间距的偏移

$[-]n \leq$  标签纸张长度 (inch或mm)



**Note: 200 DPI: 1 mm = 8 dots**

范例

一般垂直间距设定

(1) 英制系统 (inch)

GAP 0.12,0

(2) 公制系统 (mm)

GAP 3 mm,0


特殊垂直间距设定

(1) 英制系统 (inch)

GAP 0.30,-0.10

(2) 公制系统 (mm)

GAP 7.62 mm, -2.54 mm




## 4. OFFSET

该指令用于控制在剥离模式时(peel-off mode)每张卷标停止的位置，该指令仅适用于剥离模式。

### 指令语法

- (1) 英制系统 (inch)  
OFFSET m
- (2) 公制系统 (mm)  
OFFSET m mm



### 参 数

m

### 说 明

纸张停止的距离 (inch或mm)

$0 \leq m \leq 1$  (inch),  $0 \leq m \leq 25.4$  (mm)

### 范例

- (1)英制系统 (inch)  
OFFSET 0.5
- (2) 公制系统 (mm)  
OFFSET 12.7 mm

## 5. SPEED

该指令用于控制打印速度

### 指令语法

SPEED n

### 参 数

n

### 说 明

- 1.5 设定打印速度为 1.5"/sec
- 2.0 设定打印速度为 2.0"/sec
- 3.0 设定打印速度为 3.0"/sec
- 4.0 设定打印速度为 4.0"/sec

### 范例

SPEED 2.0

## 6. DENSITY

该指令用于控制打印时的浓度

指令语法

DENSITY n

参 数

n

说 明

0, 使用最淡的打印浓度

15, 使用最深的打印浓度

范例

DENSITY 7

## 7. DIRECTION

该指令用于定义打印时出纸和打印字体的方向

指令语法



DIRECTION n

参 数

n

说 明

0 或 1, 请参考图示。



## 8. REFERENCE

该指令用于定义卷标的参考坐标原点

指令语法

REFERENCE x, y

参 数

x

说 明


水平方向的坐标位置,  
以点(dot)为单位

y

垂直方向的坐标位置,  
以点(dot)为单位

范例

REFERENCE 10,10





## 9. SHIFT

该指令表示标签打印偏移量多少设置


指令语法

SHIFT n

参 数  
n

说 明

最大值为1inch，打印机为200dpi分辨率。  
(-203 < n < 203),以点为单位.



## 10. CODEPAGE

该指令用于选择所对应的国际字符集

指令语法

CODEPAGE n

参 数  
n

说 明

8-bit code page 字符集代表号

437: United States

850: Multilingual

852: Slavic

860: Portuguese

863: Canadian/French

865: Nordic

Windows code page  
1250: Central Europe  
1252: Latin I  
1253: Greek  
1254: Turkish

范例

CODEPAGE 437

## 11. CLS

该指令用于清除影像缓冲区(image buffer)的数据

指令语法

CLS

参 数

N/A

说 明

N/A

范例

CLS

## 12. FEED

该指令用于控制进纸的距离，以点(dot)表示

指令语法

FEED n

参 数

n

说 明

以点(dot)为单位  
 $1 \leq n \leq 9999$

范例

FEED 40

**注:** 200 DPI: 1 mm = 8 dots

## 13. BACKFEED&BACKUP

该指令用于控制退纸的距离，以点(dot)表示.

### 指令语法

BACKUP n  
BACKFEED n

参 数

n

说 明

以点(dot)为单位

$1 \leq n \leq 9999$

### 范例

BACKUP 40  
BACKFEED 40

**注:** 200 DPI: 1 mm = 8 dots

## 14. FORMFEED

该指令可命令打印机进一张纸

### 指令语法

FORMFEED

参 数

N/A

说 明

N/A

### 范例

FORMFEED

## 15. HOME

一般开机后打印第一张卷标时，打印位置都会不准确，打印第二张标签时打印位置就会正确，若要开机后第一张标签就要打印在正确的位置，可使用该指令

**注：使用该指令时，纸张高度需大于或等于 30 mm**

指令语法

HOME

参 数	说 明
N/A	N/A

范例

HOME

## 16. PRINT

该指令用于打印出储存于影像缓冲区内的数据


语法定义

PRINT m [,n]

参 数	说 明
m	指定打印的份数(set) $1 \leq m \leq 65535$
n	指定每份打印的张数(copy) $1 \leq n \leq 65535$

范例

```
DOWNLOAD "TEST.BAS"  
SET COUNTER @1 1  
@1="0001"  
TEXT 10,10,"3",0,1,1,@1  
PRINT 3,2  
EOP
```



## 17. SOUND

该指令用于控制蜂鸣器的频率，可设定 10 阶的声音，每阶声因的长短由第二个参数控制

### 指令语法

SOUND level, interval

### 范例

```
SOUND 5,200
SOUND 3,200
SOUND 3,200
SOUND 4,200
SOUND 2,200
SOUND 2,200
SOUND 1,200
SOUND 2,200
SOUND 3,200
SOUND 4,200
```

## 18. LIMITFEED

该指令用于设定打印机进纸时，若经过所设定的长度仍无法侦测到垂直间距，则打印机发生错误，停止进纸

### 指令语法

- (1) 英制系统 (inch)

LIMITFEED n

- (2) 公制系统 (mm)

LIMITFEED n mm

参 数

n

说 明

可使用inch或mm

### 备注

- (1). 该项设定会存于打印机内存
- (2). 当打印机初始化时，该设定值会被定为 4 inch

## 19. SELFTEST

打印机将打印出来打印机信息的标签。

指令语法

SELFTEST

<u>参 数</u>	<u>说 明</u>
N/A	N/A

范例

SELFTEST

## 卷标内容设计指令

---

### 20. BAR

在标签上画线

指令语法

BAR x, y, width, height

<u>参 数</u>	<u>说 明</u>
x	左上角水平方向起始点，以点(dot)表示
y	左上角垂直方向起始点，以点(dot)表示
width	线宽，以点(dot)表示
height	线高，以点(dot)表示

**注: 200 DPI: 1 mm = 8 dots**

范例

BAR 100, 100, 300, 200



## 21. BARCODE

该指令用来画一维条码，共计 23 种

- Code 128
- Code 128M
- EAN 128
- Interleaved 2 of 5
- Interleaved 2 of 5 with check digit
- Code 39 standard
- Code 39 full ASCII
- Code 39 full ASCII with check digit
- Code 93
- EAN 13
- EAN 13 with 2 digits add-on
- EAN 13 with 5 digits add-on
- EAN 8
- EAN 8 with 2 digits add-on
- EAN 8 with 5 digits add-on
  
- Codabar
- Postnet
- UPC-A
- UPC-A with 2 digits add-on
- UPC-A with 5 digits add-on
- UPC-E
- UPC-E with 2 digits add-on
- UPC-E with 5 digits add-on
- MSI
- PLESSEY
- China Post Code (矩阵 25 码)
- ITF14
- EAN14

### 指令语法

BARCODE X, Y, "code type", height, human readable, rotation, narrow, wide, "code"

<u>参 数</u>	<u>说 明</u>
x	条形码欲绘制的水平坐标左上角启点，以点(dot)
y	条形码欲绘制的垂直坐标左上角启点，以点(dot)
code type	
128	Code
128M	Code 128

Control code	A	B	C
096	FNC3	FNC3	NONE
097	FNC2	FNC2	NONE
098	SHIFT	SHIFT	NONE
099	CODE C	CODE C	NONE
100	CODE B	FNC4	CODE B
101	FNC4	CODE A	CODE A
102	FNC1	FNC1	FNC1
103	Start (CODE A)		
104	Start (CODE B)		
105	Start (CODE C)		

使用 “!”为条形码 subset 的控制字符，后面加三码(如上表所示)，若无指定 code 128M 的启始 subset,系统定值为 subset B .

- EAN128 Code 128,条形码 subset 采用自动选择A,B,C
- 25 Interleaved2 of 5
- 25C Interleaved2 of 5 with check digits
- 39 Auto switch full ASCII and Standard code 39 for *plus* models.
- 39C Code 39 full ASCII with check digit, Code 39 Standard with check digit ,Auto switch full ASCII and Standard code 39 for *plus* models.
- 39S Code 39 standard
- 93 Code 93
- EAN13 EAN13
- EAN13+2 EAN 13with 2 digits add-on
- EAN13+5 EAN 13with 5 digits add-on
- EAN8 EAN 8
- EAN8+2 EAN 8 with 2 digits add-on
- EAN8+5 EAN 8 with 5 digits add-on
- CODA Codabar
- POST Postnet
- ChinaPost Code 矩阵 25 码
- UPCA UPC-A
- UPCA+2 UPC-A with 2 digits add-on
- UPCA+5 UPC-A with 5 digits add-on
- UPCE UPC-E
- UPCE+2 UPC-E with 2 digits add-on
- UPCE+5 UPC-E with 5 digits add-on
- CPOST China post code
- MSI MSI code
- MSIC



- PLESSEY PLESSEY code
- ITF14 ITF14 code
- Height 条形码高度，以点(dot)表示
- human readable 0: 人眼不可识  
1: 人眼可识
- rotation 条形码旋转角度，顺时针方向
  - 0 不旋转
  - 90 顺时针方向旋转 90 度
  - 180 顺时针方向旋转 180 度
  - 270 顺时针方向旋转 270 度
- Narrow 窄 bar 宽度，以点(dot)表示
- Wide 宽 bar 宽度，以点(dot)表示

	窄比宽 1: 1	窄比宽 1: 2	窄比宽 1: 3	窄比宽 2: 5	窄比宽 3: 7
128	10x	N/A	N/A	N/A	N/A
EAN128	10x	N/A	N/A	N/A	N/A
25	N/A	10x	10x	5x	N/A
25C	N/A	10x	10x	5x	N/A
39	N/A	10x	10x	5x	N/A
39C	N/A	10x	10x	5x	N/A
93	N/A	N/A	10x	N/A	N/A
EAN13	8x	N/A	N/A	N/A	N/A
EAN13+2	8x	N/A	N/A	N/A	N/A
EAN13+5	8x	N/A	N/A	N/A	N/A
EAN8	8x	N/A	N/A	N/A	N/A
EAN8+2	8x	N/A	N/A	N/A	N/A
EAN8+5	8x	N/A	N/A	N/A	N/A
CODA	N/A	10x	10x	5x	N/A
POST	1x	N/A	N/A	N/A	N/A
UPCA	8x	N/A	N/A	N/A	N/A
UPCA+2	8x	N/A	N/A	N/A	N/A
UPCA+5	8x	N/A	N/A	N/A	N/A
UPCE	8x	N/A	N/A	N/A	N/A
UPCE+2	8x	N/A	N/A	N/A	N/A
UPCE+5	8x	N/A	N/A	N/A	N/A
CPOST	N/A	N/A	N/A	N/A	1x
MSI	N/A	N/A	10x	N/A	N/A
MSIC	N/A	N/A	N/A	N/A	10x
PLESSY	N/A	N/A	10x	N/A	N/A
ITF14	N/A	10x	10x	5x	N/A
EAN14	N/A	N/A	N/A	5x	N/A

编号的最高位数的条形码内容:

Barcode type	Maximum bar Code length	Barcode type	Maximum bar Code length
128	—	POST	5,9,11
EAN128	—	UPCA	11
25	—	UPCA + 2	13
25C	—	UPCA + 5	16
39	—	UPCE	6
39C	—	UPCE +2	8
93	—	UPCE + 5	11
EAN13	12	CPOST	—
EAN13+2	14	MSI	—
EAN13+5	17	MSIC	—
EAN8	7	PLESSY	—
EAN8+2	9	ITF14	13
EAN8+5	12	EAN14	13
COD	—		

范例

BARCODE 100,100,"39",96,1,0,2,4,"1000"

BARCODE 10,10,"128M",48,1,0,2,2,"!104!096ABCD!101EFGH"


## 22. BOX

该指令用于在卷标上绘制矩形

指令语法

BOX X\_start, Y\_start, X\_end, Y\_end,  
line thickness

(100, 100)



(200, 200)

参 数

说 明

X\_star

矩形水平方向左上角起始位置以点(dot)表示

Y\_start

矩形垂直方向左上角起始位置以点(dot)表示

X\_end

矩形水平方向右下角起始位置以点(dot)表示

Y\_end

矩形垂直方向右下角起始位置以点(dot)表示

line thickness

矩形框线厚度，以点(dot)表示

范例

BOX 100,100,200,200,5


## 23. BITMAP

该指令用于在卷标上绘制位图(非 BMP 格式图档)

指令语法

BITMAP X, Y, width, height, mode, bitmap data

参 数	说 明
x	点阵影像的水平起始位置
y	点阵影像的垂直起始位置
width	影像的宽度, 以 byte 表示
height	影像的高度, 以点(dot)表示
mode	影像绘制模式
0	OVERWRITE
1	OR
2	XOR
bitmap data	影像数据



ROW (Y- axis)	L-Byte		R-Byte	
	Binary	Hexadecimal	Binary	Hexadecimal
0	00000000	00	00000000	00
1	00000000	00	00000000	00
2	00000000	00	00000000	00
3	00000111	07	11111111	FF
4	00000011	03	11111111	FF
5	00010001	11	11111111	FF
6	00011000	18	11111111	FF
7	00011100	1C	01111111	7F
8	00011110	1E	00111111	3F
9	00011111	1F	00011111	1F
A	00011111	1F	10001111	8F
B	00011111	1F	11000111	C7
C	00011111	1F	11100011	E3
D	00011111	1F	11110111	F7
E	00011111	1F	11111111	FF
F	00011111	1F	11111111	FF

范例

SIZE 3,2  
GAP 0,0  
CLS  
BITMAP 200,200,2,16,0,  
PRINT 1,1

## 24. PUTBMP

该指令用来打印单色 PMP 格式图档

### 指令语法

PUTBMP X, Y, "filename"

<u>参 数</u>	<u>说 明</u>
X	X 方向起始点坐标, 以点(dot)表示
Y	Y 方向起始点坐标, 以点(dot)表示
Filename	欲打印的图文件名称(需已下载于打印机报存)

注: 该指令尽支持单色 BMP 图档

```
C:\BMP-PCX>DIR
Volume in drive C is WIN98
Volume Serial Number is 4140-4735

Directory of C:\BMP-PCX

06/08/2008    03:06 PM    <DIR>
06/08/2008    03:06 PM    <DIR>
06/08/2008    03:56 PM                12,430 GP.bmp
06/08/2008    03:10 PM                 1,181 GP.pcx
                2 File(s)            13,611 bytes
                2 Dir(s)  8,802,189,312 bytes free

C:\BMP-PCX>COPY CON LPT1
DOWNLOAD "GP.BMP",12430,^Z
1 file(s) copied.

C:\BMP-PCX>COPY GP.BMP/B LPT1
1 file(s) copied.

C:\BMP-PCX>COPY CON LPT1
SIZE 3,2.5
GAP 0,0
CLS
PUTBMP 100,100,"GP.BMP"
PRINT 1,1
^Z
1 file(s) copied.
C:\BMP-PCX>_
```

### 范例

PUTBMP 100,100,"LOGO.BMP"

## 25. PUTPCX

该指令用来打印单色 PCX 格式图档

### 指令语法

```
PUTPCX X, Y, "filename"
```

<u>参 数</u>	<u>说 明</u>
X	X 方向起始点坐标, 以点(dot)表示
Y	Y 方向起始点坐标, 以点(dot)表示
Filename	欲打印的图文件名称(需已下载于打印机报存)

注: 该指令尽支持单色 *PCX* 图档

```
C:\BMP-PCX>DIR
Volume in drive C is WIN98
Volume Serial Number is 4140-4735

Directory of C:\BMP-PCX

06/08/2008    03:06 PM    <DIR>
06/08/2008    03:06 PM    <DIR>
06/08/2008    03:56 PM                12,430 GP.bmp
06/08/2008    03:10 PM                1,181 GP.pcx
                2 File(s)            13,611 bytes
                2 Dir(s)   8,802,189,312 bytes free

C:\BMP-PCX>COPY CON LPT1
DOWNLOAD "GP.PCX", 12430, ^Z
1 file(s) copied.

C:\BMP-PCX>COPY GP.PCX/B LPT1
1 file(s) copied.

C:\BMP-PCX>COPY CON LPT1
SIZE 3, 2.5
GAP 0, 0
CLS
PUTBMP 100, 100, "GP.PCX"
PRINT 1, 1
^Z
1 file(s) copied.
C:\BMP-PCX>_
```

### 范例


```
PUTPCX 100,100,"LOGO.PCX"
```

## 26. ERASE

该指令用于清除影像缓冲区部份区域的数据

### 指令语法

ERASE X\_start, Y\_start, X\_width, Y\_height



#### 参 数

#### 说 明

X\_star

除区域左上角水平方向的起始点，以点(dot)表示

Y\_start

除区域左上角垂直方向的起始点，以点(dot)表示

X\_width

除区域水平方向宽度，以点(dot)表示

Y\_height

除区域垂直方向高度，以点(dot)表示

### 范例

ERASE 100,100,200,200

## 27. REVERSE

将指定的区域反相打印

### 指令语法

REVERSE X\_start, Y\_start, X\_width, Y\_height

#### 参 数

#### 说 明

X\_star

X 方向起始点，以点(dot)表示

Y\_start

Y 方向起始点，以点(dot)表示

X\_width

X 方向反白区域宽度，以点(dot)表示

Y\_height

Y 方向反白区域高度，以点(dot)表示

### 范例

REVERSE 100,100,200,200



## 28. TEXT

使用打印机 置文字(英数)打印文字

### 指令语法

TEXT X, Y, "font", rotation, x-multiplication, y-multiplication, "content"

参 数	说 明
X	文字 X 方向起始点坐标
Y	文字 Y 方向起始点坐标
Font	字体名称
1	8 x 12 dot 英数字体
2	12 x 20 dot 英数字体
3	16 x 24 dot 英数字体
4	24 x 32 dot 英数字体
5	32 x 48 dot 英数字体
6	14 x 19 dot 英数字体 OCR-B
7	21 x 27 dot 英数字体 OCR-B
8	14 x 25 dot 英数字体 OCR-A
TST24.BF2	繁体中文 24 x 24 font (大五码)
TSS24.BF2	简体中文 24 x 24 font (GB 码)
K	韩文 24 x 24 font (KS 码)
Rotation	文字旋转角度(顺时针方向)
0	0 度
90	90 度
180	180
270	270
x-multiplication:	X 方向放大倍率 1~10
y-multiplication:	Y 方向放大倍率 1~10

注： 五号字英文字母仅可打印大写字母  
若要打印双引号时(“)在程序中请使用\[“]来打印双引号  
若要打印 0D(hex)字符时，请在程序中使用\[R]来打印 CR  
若要打印 0A(hex)字符时，请在程序中使用\[A]来打印 LF

### 范例

TEXT 100,100,"4",0,1,1,"DEMO FOR TEXT"



## 询问打印机状态指令

---

### 29. <ESC>!?

询问打印机状态指令为立即响应型指令，该指令控制字符是以<ESC> (ASCII 27, escape 字符)为控制字符。即使打印机在错误状态中仍能透过 RS-232 回传一个 byte 资料来表示打印机状态，若回传值为 0 则表示打印机处于正常的状态

<u>Bit</u>	<u>状 态</u>
0	打印机未关
1	卡纸
2	缺纸
3	无碳带
4	暂停打印
5	打印中
6	机壳未关
7	错误

指令语法

<ESC>!?

参 数

无

说 明

无

### 30. <ESC>!R

该指令可有软件送出强迫打印机重新开机

指令语法

<ESC>!R

参 数

无

说 明

无

### 31. ~!@

该指令会回复打印机已打印的里程，以做为维护的参考，打印机仅回复整数部份的里程，小数的部份将被忽略，传回值将以 ASCII 字符的格式传回，以 0x0d做为结束

#### 指令语法

~!@

参 数

无

说 明

无

### 32. ~!A

该指令用于询问打印机内存大小，回传值以 10 进制字符表示，以 0x0d 做为结尾.

#### 指令语法

~!A

参 数

无

说 明

无

### 33. ~!D

该指令用于输入备份模式

#### 指令语法

~!D

参 数

无

说 明

无

### 34. ~!F

该指令用来询问打印机内存中所储存的文件名，打印机回复 ASCII 字符文件名，每个档名以 0d (hex)最为分隔，最后一个档名以 0x0d, 0x1A 做为结束。

#### 指令语法

~!F

### 35. ~!I

该指令用于询问打印机所设定的 code page code，回传格式如下

**codepage, code**

ex: 8 bit: 437, 001

有关回传讯息，请参考 **CODEPAGE** 指令

#### 指令语法

~!I

参 数

无

说 明

无

### 36. ~!T

该指令会回复打印机的型号

#### 指令语法

~!T

参 数

无

说 明

## 37. DOWNLOAD

若文件欲储存于打印机内时，文件的第一行需使用DOWNLOAD指令，当打印机收到该指令时会将数据内容储存于打印机内，需调用时，再调用其文件名称即可。DOWNLOAD 的格式可分为程序文件及数据文件，数据文件范围包含图文件、字符型文件等。

### 指令语法

#### 1. Download 程序文件

程序文件如下，打印机在收到该文件后会以 EXAMPLE.BAS 的文件名存在 存(DRAM)，EOP 代表程序结束点。

```
DOWNLOAD "EXAMPLE.BAS"  
SIZE 3,4  
GAP 0,0  
DENSITY 8  
SPEED 3  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL OFF  
CLS  
TEXT 100,100,"3",0,1,1,"EXAMPLE"  
PRINT 1  
EOP PROGRAM"
```

注：若要将程序存在打印机内，程序的第一行须加 **DOWNLOAD** “文件名.**BAS**”，程序最后一行需以 **EOP** 做结束

#### 2. Download 一资料文件

储存数据于打印机内存的格式如下

```
DOWNLOAD "FILENAME", DATA SIZE, DATA CONTENT...
```

其中

FILENAME为欲储存在打印机内存的名称，注意文件名大小写有

DATA SIZE参数为数据文件的实际大小，不包含文件头

## 38. EOP

程序结束点，需将该指令放在程序的最后一行

### 程序语法

EOP

<u>参 数</u>	<u>说 明</u>
无	无

### 范例

```
DOWNLOAD "DEMO.BAS"  
SIZE 3,4  
GAP 0,0  
DENSITY 8  
SPEED 3  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL OFF  
CLS  
TEXT 100,100,"3",0,1,1,"DEMO PROGRAM"  
FOR I=1 TO 10  
I=I+1  
NEXT  
PRINT 1  
EOP
```

## 39. FILES

该指令可打印出储存在打印机内存的文件名称，需先设定 SET DEBUG LABEL 后，该指令才有作用

### 程序语法

FILES

<u>参 数</u>	<u>说 明</u>
无	无

### 范例

请依照下列步骤列出储存于打印机内存的文件名称

```
C:\>COPY CON LPT1 <ENTER>
SET DEBUG LABEL <ENTER>
FILE<ENTER>
<CTRL>Z
C:\>
```

注: <ENTER>, <CTRL>, <CTRL>代表 PC 键盘上的相对按键

## 40. KILL

该命令用来删除储存于打印机内存的文件

### 程序语法

KILL "FILENAME"

<u>参 数</u>	<u>说 明</u>
FILENAME	删除的文件名称，注意有大小写
*	表示删除所有文件

### 范例

```
C:\>MODE COM2 96,N,8,1
C:\>COPY CON COM2
C:\>FILES
C:\>COPY CON COM2
KILL "DEMO.BAS"
^Z
C:\>FILES
```

## 41. MOVE

该指令可将存在 DRAM 的数据写到 flash memory

程序语法

MOVE

参 数

无

说 明

无

## 42. RUN

此命令是用来执行存贮在打印机内的文件

程序语法

RUN "FILENAME.BAS"

参 数

无

说 明

无

范例

```
C:\>COPY CON LPT1<ENTER>
```

```
RUN "DEMO.BAS"<ENTER>
```

```
<CTRL><Z><ENTER>
```

```
C:\>
```

## BASIC 指令及函式

---

### 43. ABS()

该函式回复整数或浮点数的绝对值

#### 程序语法

```
ABS (-100)  
ABS (-99.99)  
ABS (VARIABLE)
```

#### 范例

```
DOWNLOAD "TEST.BAS"  
SIZE 3,4  
GAP 0,0  
DENSITY 8  
SPEED 3  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL OFF  
CLS  
A=ABS(-100)  
B=ABS(-50.98)  
C=-99.99  
TEXT 100,100,"3",0,1,1,STR$(A)  
TEXT 100,150,"3",0,1,1,STR$(B)  
TEXT 100,200,"3",0,1,1,STR$(ABS(C))  
PRINT 1  
EOP
```

### 44. ASC()

该函式回传字符的 ASCII 码

#### 程序语法

```
ASC ("A")
```

#### 范例

```
DOWNLOAD "TEST.BAS"  
SIZE 3,4
```



```

GAP 0,0
DENSITY 8
SPEED 3
DIRECTION 0
REFERENCE 0,0
SET PEEL OFF
CLS
CODE1=ASC("A")
TEXT 100,100,"3",0,1,1,STR$(CODE1)
PRINT 1
EOP

```

## 45. CHR\$( )

该函数传回所指定的 ASCII code 字符

### 程序语法

CHR\$(n)

参 数

无

说 明

无

### 范例

```

DOWNLOAD "TEST.BAS"
SIZE 3,4
GAP 0,0
DENSITY 8
SPEED 3
DIRECTION 0
REFERENCE 0,0
SET PEEL OFF
CLS
A=65
WORD$=CHR$(A)
TEXT 100,100,"3",0,1,1,WORD$
PRINT 1
EOP

```

## 46. END

主程序的结束点

程序语法

END

范例

```
DOWNLOAD "TEST.BAS"  
SIZE 3,4  
GAP 0,0  
DENSITY 8  
SPEED 3  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL OFF  
CLS  
A=1  
:START  
IF A>10 THEN END ELSE A=A+1  
GOTO START  
TEXT 100,100,"3",0,1,1,STR$(A)  
PRINT 1  
EOP
```

## 47. EOF()

该命令用来判断文件指针是否已到文件的结束位置

程序语法

EOF (File Handle)

参 数

File handle

说 明

0 或 1

参 数

非零

0

说 明

表示指针已到文件最后结束位置

指针未到达档尾

## 范例

```
DOWNLOAD "DEMO.BAS"
SIZE 3,4
GAP 0,0
DENSITY 8
SPEED 3
DIRECTION 0
REFERENCE 0,0
SET PEEL OFF
OPEN "DATA",0
SEEK 0,4
:A
ITEM$=""
READ 0,ITEM$,PRICE,QTY
TEXT 10,10,"3",0,1,1,"SEEK          COMMAND TEST (SHIFT 4 CHARACTERS)"
BOX 98,48,502,514,2
A$="ITEMS:"+ITEM$
B$="PRICE:"+STR$(PRICE)
C$="QTY:"+STR$(QTY)
TEXT 128,114,"2",0,1,1,A$
TEXT 130,198,"2",0,1,1,B$
TEXT 132,268,"2",0,1,1,C$
BARCODE 132,365,"39",96,1,0,2,4,"PRICE-2000"
PRINT 1
I=EOF(0)
IF I=0 THEN GOTO A
PRINT 1
EOP
```

## 48. OPEN

该指令用于开启储存于打印机内存的文件，打印机最多能同时一次开启两个文件，使用该指令时文件需已储存于打印机内。

### 程序语法

OPEN "Filename", File handle

<u>参 数</u>	<u>说 明</u>
Filename	储存于内存的文件名称
Filename	0 or 1

## 范例

下载文件的名称和数据格式如下：

该文件的格式包括：

```
DOWNLOAD "DATA",20,Computer<CR>
20000<CR>
15<CR>
```

综合上述内容的文件数据下载，文件名“数据”。按照以下步骤下载数据到打印机

```
<under MS-DOS mode>:
C:\>MODE COM2:96,N,8,1 □
C:\>COPY DATA /B COM2 □
```

上面的例子中设置如下：波特率 9600，奇偶校验无，8 位数据，停止位 1。如果一个文件名为“DEMO.BAS”，就是要下载的文件格式包括：

```
DOWNLOAD "DEMO.BAS"
SIZE 3.00,3.00
CLS
SPEED 2
DENSITY 8
SET PEEL OFF
DIRECTIO 0
REFERENCE 0,0
OPEN "DATA",1
SEEK 1,0
READ 1,ITEM$,PRICE,QTY
I=EOF(1)
IF I>0 THEN END
BOX 98,48,502,514,2
A$="ITEMS:"+ITEM$
B$="PRICE:"+STR$(PRICE)
C$="QTY:"+STR$(QTY)
TEXT 128,114,"2",0,1,1,A$
TEXT 130,198,"2",0,1,1,B$
TEXT 132,268,"2",0,1,1,C$
BARCODE 132,365,"39",96,1,0,2,4,"PRICE-2000"
PRINT 1
CLOSE 1
EOP
```

上面的例子是打印机打开文件里“数据”的文件处理和阅读项目文件。

## 49. READ

该指令用于读取已存于打印机内存的文件

### 程序语法

READ file handle, variables

<u>参 数</u>	<u>说 明</u>
file handle	0 or 1
variables	字符串、整数或浮点变量

### 范例

```
DOWNLOAD "OPEN1.BAS"
SIZE 3,4
GAP 0,0
DENSITY 8
SPEED 3
DIRECTION 0
REFERENCE 0,0
SET PEEL OFF
CLS
OPEN "DATA",0
OPEN "DATA1",1
SEEK 0,4
:A
ITEM$=""
READ 0,ITEM$,PRICE,QTY
TEXT 10,10,"3",0,1,1,"SEEK          COMMAND TEST (SHIFT 4
CHARACTERS)"
BOX 98,48,502,514,2
A$="ITEMS:"+ITEM$
B$="PRICE:"+STR$(PRICE)
C$="QTY:"+STR$(QTY)
TEXT 128,114,"2",0,1,1,A$
TEXT 130,198,"2",0,1,1,B$
TEXT 132,268,"2",0,1,1,C$
BARCODE 132,365,"39",96,1,0,2,4,"PRICE-2000"
PRINT 1
I=EOF(0)
IF I=0 THEN GOTO A
SEEK 1,0
:B
```

```

READ 1,ITEM$,PRICE,QTY
TEXT 10,10,"4",0,1,1,"OPEN, READ, EOF() COMMAND TEST"
BOX 98,48,502,514,2
A$="ITEMS:"+ITEM$
B$="PRICE:"+STR$(PRICE)
C$="QTY:"+STR$(QTY)
TEXT 128,114,"2",0,1,1,A$
TEXT 130,198,"2",0,1,1,B$
TEXT 132,268,"2",0,1,1,C$
BARCODE 132,365,"39",96,1,0,2,4,"PRICE-2000"
PRINT 1
I=EOF(1)
IF I=0 THEN GOTO B
OPEN "DATA2",0
CLS
Z$=""
Z$=FREAD$(0,6)
TEXT 10,20,"4",0,1,1,"FREAD$() FUNCTION TEST"
TEXT 10,70,"4",0,1,1,"ITEM3$= "+Z$
J=LOF("DATA2")
TEXT 10,140,"3",0,1,1,"THE FILE SIZE OF DATA2 IS: "+STR$(J)+"
Bytes"
PRINT 1
EOP

```

## 50. SEEK

该指令用来移动文件指针到某一特定的位置.

### 程序语法

SEEK file handle, offset

<u>参 数</u>	<u>说 明</u>
file handle	0 or 1 文件指标的偏移量

### 范例

```

DOWNLOAD "TEST.BAS"
SIZE 3,4
GAP 0,0
DENSITY 8
SPEED 3

```

```

DIRECTION 0
REFERENCE 0,0
SET PEEL OFF
CLS
OPEN "DATA",0
SEEK 0,4
ITEM$=""
READ 0,ITEM$,PRICE,QTY
TEXT 10,10,"3",0,1,1,"SEEK COMMAND TEST (SHIFT 4
CHARACTERS)"
BOX 98,48,502,514,2
A$="ITEMS:"+ITEM$
B$="PRICE:"+STR$(PRICE)
C$="QTY:"+STR$(QTY)
TEXT 128,114,"2",0,1,1,A$
TEXT 130,198,"2",0,1,1,B$
TEXT 132,268,"2",0,1,1,C$
BARCODE 132,365,"39",96,1,0,2,4,"PRICE-2000"
PRINT 1
EOP

```

## 51. LOF()

该指令可回传已打开文件的文件大小，以 byte 表示

### 程序语法

LOF ("FILENAME")

参 数

Filename

说 明

已下载在打印机内存的文件名称

### 范例

```

DOWNLOAD "OPEN2.BAS"
SIZE 3,4
GAP 0,0
DENSITY 8
SPEED 3
DIRECTION 0
REFERENCE 0,0
SET PEEL OFF
CLS
OPEN "DATA2",0

```

```

CLS
Z$=""
Z$=FREAD$(0,6)
TEXT 10,20,"4",0,1,1,"FREAD$() FUNCTION TEST"
TEXT 10,70,"4",0,1,1,"ITEM3$= "+Z$
J=LOF("DATA2")
TEXT 10,140,"3",0,1,1,"THE FILE SIZE OF DATA2 IS: "+STR$(J)+"
Bytes"
PRINT 1
EOP

```

## 52. FREAD\$()

该指令用于读取已打开文件内所指定 byte 数的数据

### 程序语法

FREAD\$ (file handle, byte)

参 数	说 明
File handle	0 或 1
Byte	欲读取数据的byte数

### 范例

```

DOWNLOAD "FREAD.BAS"
SIZE 3,4
GAP 0,0
DENSITY 8
SPEED 3
DIRECTION 0
REFERENCE 0,0
SET PEEL OFF
CLS
OPEN "DATA2",0
CLS
Z$=""
Z$=FREAD$(0,6)
TEXT 10,20,"4",0,1,1,"FREAD$() FUNCTION TEST"
TEXT 10,70,"4",0,1,1,"ITEM3$= "+Z$
J=LOF("DATA2")
TEXT 10,140,"3",0,1,1,"THE FILE SIZE OF DATA2 IS: "+STR$(J)+"
Bytes"
PRINT 1
EOP

```



## 53. FOR...NEXT LOOP

循环指令可自动执行的程序，直到条件满足为止。请勿由循环外部直接跳到循环内部执行，否则将发生无法预期的错误。

### 程序语法

```
For variable = start TO end STEP increment  
    程序  
NEXT
```

<u>参 数</u>	<u>说 明</u>
Variable	变量名称最多可达8个字符
start	可为整数或浮点数
end	可为整数或浮点数
increment	可为整数或浮点数，正数或负数

### 范例

```
DOWNLOAD "FREAD.BAS"
```

```
SIZE 3,4
```

```
GAP 0,0
```

```
DENSITY 8
```

```
SPEED 3
```

```
DIRECTION 0
```

```
REFERENCE 0,0
```

```
SET PEEL OFF
```

```
CLS
```

```
A$=""
```

```
B$=""
```

```
C$=""
```

```
H$=""
```

```
FOR I=1 TO 10 STEP 1
```

```
A$=A$+STR$(I)+" "
```

```
TEXT 10,10,"3",0,1,1,A$
```

```
NEXT
```

```
FOR I=1 TO 1000 STEP 100
```

```
B$=B$+STR$(I)+" "
```

```
TEXT 10,50,"3",0,1,1,B$
```

```
NEXT
```

```
FOR I=50 TO 10 STEP -10
```

```
C$=C$+STR$(I)+" "
```

```
TEXT 10,100,"3",0,1,1,C$
```

```

NEXT
FOR I=1 TO 5 STEP 0.5
HS=HS+STR$(I)+" "
TEXT 10,150,"3",0,1,1,HS
NEXT
PRINT 1
EOP

```

## 54. IF...THEN...ELSE

条件判断式指令

程序语法

IF condition THEN statement [ ELSE statement ]

IF...THEN...ELSE 指令需写在同一行，不可分两行写，程序长度不可大于255 个字符

参 数	说 明
condition	可使用的运算符有<, >, =
statement	程序仅能用一行

范例

```

DOWNLOAD "IFTHEN.BAS"
SIZE 3,4
GAP 0,0
DENSITY 8
SPEED 3
DIRECTION 0
REFERENCE 0,0
SET PEEL OFF
CLS
A=50
B=5
C$=""
D$=""

:L1
IF A>100 THEN GOTO L1 ELSE A=A+10
C$=STR$(A)+" IS SMALLER THAN 100"
TEXT 100,10,"4",0,1,1,C$
PRINT 1

```

```

END
:L2
A=A+B
D$=STR$(A)+" IS LARGER THAN 100"
TEXT 100,100,"4",0,1,1,D$
PRINT 1
GOTO L1
EOP

```

## 55. GOSUB...RETURN

该指令可由主程序跳到子程序执行再跳回主程序

### 程序语法

```

GOSUB LABEL
      statement
END
:LABEL
      statement
RETURN

```

参 数

LABEL

说 明

子程序的起始点，LABEL名称不可超过 8 个字符

### 范例

```

DOWNLOAD "GOSUB1.BAS"
SIZE 3,4
GAP 0,0
DENSITY 8
SPEED 3
DIRECTION 0
REFERENCE 0,0
SET PEEL OFF
CLS
TEXT 10,600,"5",0,1,1,"GOSUB & RETURN COMMAND TEST"
X=300
Y=300
GOSUB DASH
GOSUB DR_LINE
PRINT 1
END

```

```

:DR_LINE
BOX X,Y,X+200,Y+200,5
RETURN

:DASH
SET PEEL OFF
DENSITY 13
FOR I=21 TO 50
TEXT 10,I,100,"3",0,1,1,"=====
NEXT
RETURN
EOP

```

## 56. GOTO

此命令是用来分行指定的标签。标签不能超过 8 个字符。

### 程序语法

```
GOTO LABEL
```

```
:LABEL
```

<u>参 数</u>	<u>说 明</u>
无	无

### 范例

```

DOWNLOAD "GOSUB1.BAS"
SIZE 3,4
GAP 0,0
DENSITY 8
SPEED 3
DIRECTION 0
REFERENCE 0,0
SET PEEL OFF
CLS
A=0
TOTAL=0
:START
IF A<100 THEN GOTO SUM ELSE GOTO PRTOUIT
:SUM
A=A+1
TOTAL=TOTAL+A
GOTO START

```

```
:PRTOUT
B$="THE SUMMATION OF 1..100 IS "+STR$(TOTAL)
TEXT 10,100,"3",0,1,1,B$
PRINT 1
END
EOP
```

## 57. REM

该指令用于程序批注

### 程序语法

REM

### 范例

```
REM *****
REM This is a demonstration program*
REM *****
DOWNLOAD "REMARK.BAS"
SIZE 3,4
GAP 0,0
DENSITY 8
SPEED 3
DIRECTION 0
REFERENCE 0,0
SET PEEL OFF
CLS
TEXT 50,50,"3",0,1,1,"REMARK DEMO PROGRAM"
EOP
```

## 58. INT()

该函数将传回浮点数的整数部份

### 程序语法

INT (n)

<u>参 数</u>	<u>说 明</u>
n	n 可以是正数或负数、浮点数或数学表达式

### 范例

```
DOWNLOAD "DEMO5.BAS"  
SIZE 3,4  
GAP 0,0  
DENSITY 8  
SPEED 3  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL OFF  
CLS  
A=INT(99.99)  
B=INT(-199.89)  
C=INT(80)  
TEXT 50,100,"3",0,1,1,"INT(99.99)" + STR$(A)  
TEXT 50,150,"3",0,1,1,"INT(-199.89)" + STR$(B)  
TEXT 50,200,"3",0,1,1,"INT(80)" + STR$(C)  
PRINT 1  
EOP
```

## 59. LEFT\$( )

该函数传回字符串中最左边指定的字符

### 程序语法

LEFT\$(X\$, n)

<u>参 数</u>	<u>说 明</u>
X\$	欲处理的字符串
n	欲截取回传的字符数

## 范例

```
DOWNLOAD "STR1.BAS"  
SIZE 3.00,4.00  
GAP 0.12,0.00  
SPEED 2.0  
DENSITY 8  
DIRECTION 0  
REFERENCE 0,0  
SET DEBUG LABEL  
CLS  
A$="TAIWAN SEMICONDUCTOR CO., LTD"  
C$=LEFT$(A$,10)  
TEXT 10,10,"3",0,1,1,A$  
TEXT 10,100,"3",0,1,1,"10 LEFT CHARS "+C$  
PRINT 1  
EOP
```

## 60. LEN()

该函数返回字符串的长度

### 程序语法

LEN (string)

参 数

string

说 明

欲量测的字符串

## 范例

```
DOWNLOAD "DEMO6.BAS"  
SIZE 3.00,4.00  
GAP 0.12,0.00  
SPEED 2.0  
DENSITY 8  
DIRECTION 0  
REFERENCE 0,0  
SET DEBUG LABEL  
CLS  
A$="TAIWAN SEMICONDUCTOR CO., LTD"  
B=LEN(A$)  
TEXT 10,10,"3",0,1,1,A$  
TEXT 10,50,"3",0,1,1,"STRING LENGTH="+STR$(B)  
PRINT 1  
EOP
```

## 61. MID\$( )

该函式用来传回字符串中某几个字符

### 程序语法

MID\$(string, m, n)

<u>参 数</u>	<u>说 明</u>
string	欲处理的字符串
m	字符串中第 m 个位置起始位置 $1 \leq m \leq$ 字符串长度
n	欲传回的字符数

### 范例

```
DOWNLOAD "DEMO7.BAS"  
SIZE 3.00,4.00  
GAP 0.12,0.00  
SPEED 2.0  
DENSITY 8  
DIRECTION 0  
REFERENCE 0,0  
SET DEBUG LABEL  
CLS  
A$="TAIWAN SEMICONDUCTOR CO., LTD"  
E$=MID$(A$,11,10)  
TEXT 10,10,"3",0,1,1,A$  
TEXT 10,200,"3",0,1,1,"10 MIDDLE CHARS "+E$  
PRINT 1  
EOP
```

## 62. RIGHT\$( )

该函式将从字符串的最右边传回 n 个字符

### 程序语法

RIGHT\$(X\$, n)

<u>参 数</u>	<u>说 明</u>
X\$	欲处理的字符串
n	从字符串的最右边传回 n 个字符



范例

```
DOWNLOAD "DEMO8.BAS"  
SIZE 3.00,4.00  
GAP 0.12,0.00  
SPEED 2.0  
DENSITY 8  
DIRECTION 0  
REFERENCE 0,0  
SET DEBUG LABEL  
CLS  
A$="TAIWAN SEMICONDUCTOR CO., LTD"  
D$=RIGHT$(A$,10)  
TEXT 10,10,"3",0,1,1,A$  
TEXT 10,150,"3",0,1,1,"10 RIGHT CHARS "+D$  
PRINT 1  
EOP
```

### 63. STR\$( )

该函式将数字转换为字符串

程序语法

STR\$(n)

范例

```
DOWNLOAD "DEMO8.BAS"  
SIZE 3.00,4.00  
GAP 0.12,0.00  
SPEED 2.0  
DENSITY 8  
DIRECTION 0  
REFERENCE 0,0  
SET DEBUG LABEL  
CLS  
A$="TAIWAN SEMICONDUCTOR CO., LTD"  
D$=RIGHT$(A$,10)  
TEXT 10,10,"3",0,1,1,A$  
TEXT 10,150,"3",0,1,1,"10 RIGHT CHARS "+D$  
PRINT 1  
EOP
```

## 64. VAL()

该函数将数字字符串转为数字型态

### 程序语法

VAL ("numeric character")

<u>参 数</u>	<u>说 明</u>
numeric character	“0~9”, “.”

### 范例

```
DOWNLOAD "DEMO10.BAS"  
SIZE 3.00,4.00  
GAP 0.12,0.00  
SPEED 2.0  
DENSITY 8  
  
DIRECTION 0  
REFERENCE 0,0  
SET DEBUG LABEL  
CLS  
A$="TAIWAN SEMICONDUCTOR CO., LTD"  
F$="100"  
G$="500"  
H=VAL(F$)+VAL(G$)  
I$=STR$(H)  
TEXT 10,10,"3",0,1,1,A$  
TEXT 10,250,"3",0,1,1,"VAL FUNCTION TEST="+I$  
PRINT 1  
EOP
```

## 65. BEEP

该指令为控制蜂鸣器收到该指令时会发出一声响

### 程序语法

BEEP

<u>参 数</u>	<u>说 明</u>
无	无

### 范例

```
DOWNLOAD "DEMO11.BAS"  
SIZE 3,4  
GAP 0,0  
DENSITY 8  
SPEED 3  
DIRECTION 0  
REFERENCE 0,0  
  
SET PEEL OFF  
CLS  
:START  
CLS  
A$=""  
TEXT 20,50,"3",0,1,1,"Please connect LCD keyboard for  
testing"  
BEEP  
INPUT "Enter the code number"; A$  
BARCODE 20,100,"39",48,1,0,2,5,A$  
PRINT 1  
GOTO START  
EOP
```

# 打印机外围功能设定指令

---

## 66. SET COUNTER

设定计数器及增量，该指令不支持数学表达式

### 程序语法

SET COUNTER @n step

<u>参 数</u>	<u>说 明</u>
@n	n:计数器号码，打印机可使用50组计数器(@0~@49)
step	计数器跳号的增量，可为正或负数 -999999999<= step <=999999999

### 范例

```
DOWNLOAD "DEMO13.BAS"  
SIZE 3,4  
GAP 0,0  
DENSITY 8  
SPEED 3  
DIRECTION 0  
REFERENCE 0,0  
  
SET PEEL OFF  
SET COUNTER @0 1  
SET COUNTER @1 5  
CLS  
@1="00001"  
@2="TSC00001"  
TEXT 50,50,"3",0,1,1,@1  
BARCODE 50,500,"39",48,1,0,2,4,@2  
PRINT 1  
EOP
```

## 67. SET KEY1, SET KEY2

该指令用来启动/关闭 KEY1, KEY2 的预设功能。

### 程序语法

```
SET KYE1 ON /OFF  
SET KEY2 ON /OFF
```

<u>参 数</u>	<u>说 明</u>
ON	启动 KEY1 为暂停功能 启动 KEY2 为进纸功能
OFF	关闭 KEY1 预设为暂停功能 关闭 KEY2 预设为进纸功能

**注：** 关闭电源时该设定值任会储存在打印机内。

### 范例

```
DOWNLOAD "DEMO17.BAS"  
SIZE 3,4  
GAP 0,0  
DENSITY 8  
SPEED 3  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL OFF  
SET KEY1 OFF  
CLS  
:START  
A=GETKEY()  
IF A=0 THEN GOTO PAUSEB  
IF A=1 THEN GOTO FEEDB  
:PAUSEB  
CLS  
TEXT 50,10,"4",0,1,1,"PAUSE key is pressed!"  
PRINT 1  
GOTO START  
:FEEDB  
CLS  
TEXT 50,10,"4",0,1,1,"FEED key is pressed!"  
PRINT 1  
GOTO START  
EOP
```

## 68. SET PEEL

该指令用来启动/关闭剥离模式默认值为关闭

### 程序语法

SET PEEL ON/OFF

<u>参 数</u>	<u>说 明</u>
ON	启动剥离模式
OFF	关闭剥离模式

### 范例

```
REM SELF-PEELING FUNCTION ON
SIZE 3,4
GAP 0,0
DENSITY 8
SPEED 3
DIRECTION 0EFERENCE 0,0
SET PEEL ON
SET KEY1 OFF
CLS
TEXT 50,100,"3",0,1,1,"SELF-PEELING FUNCTION TEST"
PRINT
```

## 69. SET TEAR & SET STRIPPER

此命令是用来启用/禁用撕纸位置走到撕纸处,此设置关掉电源后将保存在打印机内。

### 程序语法

```
SET TEAR ON/OFF
SET STRIPPER ON/OFF
```

<u>参 数</u>	<u>说 明</u>
ON	启用撕纸位置走到撕纸处
OFF	禁用撕纸位置走到撕纸处,该命在起始位置有效。

## 范例

```
REM ***TEAE FUNCTION ON***  
SIZE 3,3  
  
GAP 0.08,0  
DENSITY 8  
SPEED 4  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL OFF  
SET TEAR ON  
CLS  
TEXT 50,100,"3",0,1,1,"TEAR FUNCTION TEST"  
PRINT 1
```

## 70. SET HEAD

此设置用于启用/禁用打印头合盖传感器。如果合盖传感器是开放的打印头被打开，没有任何讯息传回。此设置将保存在打印机内存。

### 程序语法

SET HEAD ON/OFF

<u>参 数</u>	<u>说 明</u>
ON	启用打印头合盖传感器
OFF	禁用打印头合盖传感器

### 范例

```
SET HEAD ON  
SET HEAD OFF
```

## 71. SET COM1

该指令用来设定打印机串口的通讯参数

### 程序语法

SET COM1 baud, parity, data, stop

<u>参 数</u>	<u>说 明</u>
baud	可使用的波特率如下: 24: 2400 bps 48: 4800 bps 96: 9600 bps 19: 19200 bps
Parity	parity N: None parity check E: Even parity check O: Odd parity check
Data	Data bit 8: 8 bits data 7: 7 bits data
Stop	Stop bit 1: 1 stop bit 2: 2 stop bits

### 范例

```
SIZE 3,4  
GAP 0,0  
DENSITY 8  
SPEED 3  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL ON  
SET KEY1 OFF  
SET DEBUG LABEL  
SET RIBBON OFF  
SET COM1 96,N,8,1  
CLS  
BARCODE 100,100,"39",48,1,0,2,5,"CODE 39"  
PRINT 1
```



## 72. SET PRINTKEY

此命令将打印一个标签并走到下一个标签的间隙到撕纸位置处，按下 FEED 按键，打印下一个标签或多份的标签。如果标签内容包括串行文字或条形码，它将改变序号。此设置将保存在打印机内存。

### 程序语法

SET PRINTKEY OFF/ON/AUTO/<num>

参 数	说 明
OFF	禁止此功能
ON	开启此功能
AUTO	自动开启此功能
<num>	按FEED键来按多少下

### 范例

```
Execute
SIZE 3,2,5
GAP 0.12,0
SET PRINTKEY ON
SET COUNTER @0 1
@0="0001"
CLS
TEXT 10,10,"5",0,1,1,@0
PRINT 1
```

### Execute:

Syntax	Receive "PRINT m"	Print Out
SET PRINTKEY ON or SET PRINTKEY AUTO	1.) PRINT 2	Label 1~2
	2.) Press FEED key	Label 3~4

Syntax	Receive "PRINT m,n"	Print Out
SET PRINTKEY ON or SET PRINTKEY AUTO	1.) PRINT 1,2	Label 1, Label 1
	2.) Press FEED key	Label 2, Label 2

Syntax	Receive "PRINT -1,n"	Print Out
SET PRINTKEY ON or SET PRINTKEY AUTO	1.) PRINT -1,2	Label 1, Label 1
	2.) Press FEED key	Label 1, Label 1

Syntax	Receive "PRINT m"	Print Out
SET PRINTKEY 5	1.) PRINT 2	Label 1~2
	2.) Press FEED key	Label 3~7

Syntax	Receive "PRINT m,n"	Print Out
SET PRINTKEY 5	1.) PRINT 1,2	Label 1, Label 1
	2.) Press FEED key	Label 2~6

Syntax	Receive "PRINT -1,n"	Print Out
SET PRINTKEY 5	1.) PRINT -1,2	Label 1, Label 1
	2.) Press FEED key	Label 1, Label 1

### 73. SET REPRINT

此命令将禁用/启用标签机在无纸或开盖错误发生后，上纸或合盖后重新打印一次标签内容。

#### 程序语法

SET REPRINT OFF/ON

<u>参 数</u>	<u>说 明</u>
OFF	禁止此功能
ON	启用此功能

#### 范例

SET REPRINT ON

## 74. PEEL

此命令是用来获取纸存在剥离传感器。其属性是只读。

### 程序语法

PEEL

<u>参 数</u>	<u>说 明</u>
0	当没有纸在纸存在传感器上方时返回值
1	当有纸在纸存在传感器上方时返回值

### 范例

```
DOWNLOAD "DEMO19.BAS"  
SIZE 4,4  
GAP 0,0  
DENSITY 8  
SPEED 3  
DIRECTION 0  
REFERENCE 0,0  
SET CUTTER 1  
SET PEEL OFF  
SET LED1 OFF  
CLS  
IF PEEL=1 THEN LED1=1  
EOP
```

## 75. KEY1,KEY2

此命令用来读取打印机按键的状态.

### 程序语法

KEYm=n

<u>Key</u>	<u>Return Value</u>
KEY1(PAUSE)	0:released 1:pressed
KEY2(FEED)	0:released 1:pressed

### 范例

```
DOWNLOAD "DEMO.BAS" SIZE 3,1
GAP 0,0
SPEED 4
DENSITY 8
DIRECTION 1
REFERENCE 0,0
SET KEY1 OFF
:START
IF KEY1=1 THEN
  CLS
  TEXT 100,10,"3",0,1,1,"KEY FUNCTION TEST"
  PRINT 1,1
GOTO START
EOP
DEMO
```

# 打印机全域变数

---

## 76. @LABEL

该变量记录打印机已打印标签张数

### 程序语法

Write attribute: @LABEL=n

Read attribute: A=@LABEL

参 数	说 明
n	打印标签张数 0<n<65535

### 范例

```
DOWNLOAD "DEMO20.BAS"  
SIZE 3,4  
GAP 0,0  
DENSITY 8  
SPEED 3  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL ON  
SET KEY1 OFF  
SET DEBUG LABEL  
SET RIBBON OFF  
SET COM1 96,N,8,1  
CLS  
IF @LABEL=100 THEN @LABEL=0 ELSE TEXT  
100,100,"3",0,1,1,STR$(@LABEL)  
PRINT 1  
EOP
```